NAG Fortran Library Routine Document

F08PNF (ZGEES)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08PNF (ZGEES) computes the eigenvalues, the Schur form T, and, optionally, the matrix of Schur vectors Z for an n by n complex nonsymmetric matrix A.

2 Specification

```
SUBROUTINE F08PNF (JOBVS, SORT, SELECT, N, A, LDA, SDIM, W, VS, LDVS,
1                  WORK, LWORK, RWORK, BWORK, INFO)
INTEGER            N, LDA, SDIM, LDVS, LWORK, INFO
double precision   RWORK(*)
complex*16         A(LDA,*), W(*), VS(LDVS,*), WORK(*)
LOGICAL            SELECT, BWORK(*)
CHARACTER*1        JOBVS, SORT
EXTERNAL           SELECT
```

The routine may be called by its LAPACK name ***zgees***.

3 Description

The Schur factorization of A is given by

$$A = ZTZ^H,$$

where Z is orthogonal, the matrix of Schur vectors, and T is upper triangular.

Optionally, F08PNF (ZGEES) also orders the eigenvalues on the diagonal of the Schur form so that selected eigenvalues are at the top left. The leading columns of Z then form an orthonormal basis for the invariant subspace corresponding to the selected eigenvalues.

A complex matrix is in Schur form if it is upper triangular.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: JOBVS – CHARACTER*1 *Input*

On entry: if JOBVS = 'N', Schur vectors are not computed.

If JOBVS = 'V', Schur vectors are computed.

2: SORT – CHARACTER*1 *Input*

On entry: specifies whether or not to order the eigenvalues on the diagonal of the Schur form:

if SORT = 'N', eigenvalues are not ordered;
if SORT = 'S', eigenvalues are ordered (see SELECT).

3: SELECT – LOGICAL FUNCTION, supplied by the user. *External Procedure*

If SORT = 'S', SELECT is used to select eigenvalues to sort to the top left of the Schur form.

If SORT = 'N', SELECT is not referenced and F08PNF (ZGEES) may be called with the dummy function F08PNZ.

Its specification is:

```
LOGICAL FUNCTION SELECT (W)
complex*16                 W
```

1: W – ***complex*16*** *Input*

On entry: the eigenvalue $\mathrm{W}(j)$ is selected if $\mathrm{SELECT}(\mathrm{W}(j))$ is .TRUE..

SELECT must be declared as EXTERNAL in the (sub)program from which F08PNF (ZGEES) is called. Parameters denoted as *Input* must **not** be changed by this procedure.

4: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

5: A(LDA,$*$) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the n by n matrix A.

On exit: has been overwritten by its Schur form T.

6: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08PNF (ZGEES) is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{N})$.

7: SDIM – INTEGER *Output*

On exit: if SORT = 'N', SDIM = 0.

If SORT = 'S', SDIM = number of eigenvalues for which SELECT is true.

8: W($*$) – ***complex*16*** array *Output*

Note: the dimension of the array W must be at least $\max(1, \mathrm{N})$.

On exit: contains the computed eigenvalues, in the same order that they appear on the diagonal of the output Schur form T.

9: VS(LDVS,$*$) – ***complex*16*** array *Output*

Note: the second dimension of the array VS must be at least $\max(1, \mathrm{N})$ if JOBVS = 'V' and at least 1 otherwise.

On exit: if JOBVS = 'V', VS contains the unitary matrix Z of Schur vectors.

If JOBVS = 'N', VS is not referenced.

10: LDVS – INTEGER *Input*

On entry: the first dimension of the array VS as declared in the (sub)program from which F08PNF (ZGEES) is called.

Constraints:

if JOBVS = 'V', LDVS $\geq \max(1, \mathrm{N})$;
LDVS ≥ 1 otherwise.

11: WORK$(*)$ – ***complex*16*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, \mathrm{LWORK})$.

On exit: if INFO $= 0$, WORK(1) returns the optimal LWORK.

12: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08PNF (ZGEES) is called.

For good performance, LWORK must generally be larger than the minimum, say $2 \times \mathrm{N} + nb \times \mathrm{N}$, where nb is the optimal block size for F08NSF (ZGEHRD).

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Constraint: LWORK $\geq \max(1, 2 \times \mathrm{N})$.

13: RWORK$(*)$ – ***double precision*** array *Workspace*

Note: the dimension of the array RWORK must be at least $\max(1, \mathrm{N})$.

14: BWORK$(*)$ – LOGICAL array *Workspace*

Note: the dimension of the array BWORK must be at least 1 if SORT = 'N' and at least $\max(1, \mathrm{N})$ otherwise.

If SORT = 'N', BWORK is not referenced.

15: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value.

INFO > 0

If INFO $= i$ and $i \leq \mathrm{N}$, the QR algorithm failed to compute all the eigenvalues

INFO $= \mathrm{N} + 1$

The eigenvalues could not be reordered because some eigenvalues were too close to separate (the problem is very ill-conditioned).

INFO $= \mathrm{N} + 2$

After reordering, roundoff changed values of some complex eigenvalues so that leading eigenvalues in the Schur form no longer satisfy SELECT = .TRUE.. This could also be caused by underflow due to scaling.

7 Accuracy

The computed Schur factorization satisfies

$$A + E = ZTZ^H,$$

where

$$\|E\|_2 = \mathrm{O}(\epsilon)\|A\|_2,$$

and ϵ is the ***machine precision***. See Section 4.8 of Anderson *et al.* (1999) for further details.

8 Further Comments

The total number of floating-point operations is proportional to n^3.

The real analogue of this routine is F08PAF (DGEES).

9 Example

To find the Schur factorization of the matrix

$$A = \begin{pmatrix} -3.97 - 5.04i & -4.11 + 3.70i & -0.34 + 1.01i & 1.29 - 0.86i \\ 0.34 - 1.50i & 1.52 - 0.43i & 1.88 - 5.38i & 3.36 + 0.65i \\ 3.31 - 3.85i & 2.50 + 3.45i & 0.88 - 1.08i & 0.64 - 1.48i \\ -1.10 + 0.82i & 1.81 - 1.59i & 3.25 + 1.33i & 1.57 - 3.44i \end{pmatrix}.$$

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08PNF Example Program Text
*     Mark 21.  NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NB, NMAX
      PARAMETER        (NB=64,NMAX=10)
      INTEGER          LDA, LDVS, LWORK
      PARAMETER        (LDA=NMAX,LDVS=NMAX,LWORK=NMAX+NMAX*NB)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, LWKOPT, N, SDIM
*     .. Local Arrays ..
      COMPLEX *16      A(LDA,NMAX), VS(LDVS,NMAX), W(NMAX), WORK(LWORK)
      DOUBLE PRECISION RWORK(NMAX)
      LOGICAL          DUMMY(1)
      CHARACTER        CLABS(1), RLABS(1)
*     .. External Functions ..
      LOGICAL          F08PNZ
      EXTERNAL         F08PNZ
*     .. External Subroutines ..
      EXTERNAL         X04DBF, ZGEES
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08PNF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read the matrix A from data file
*
         READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
```

```
*
*        Find the Schur factorization
*
         CALL ZGEES('Vectors (Schur)','No sort',F08PNZ,N,A,LDA,SDIM,W,
     +             VS,LDVS,WORK,LWORK,RWORK,DUMMY,INFO)
         LWKOPT = WORK(1)
*
         IF (INFO.GT.0) THEN
            WRITE (NOUT,99999) 'Failure in ZGEES. INFO =', INFO
         ELSE
*
*           Print out factors of the Schur factorization
*
            IFAIL = 0
            CALL X04DBF('General',' ',N,N,A,LDA,'Bracketed','F7.3',
     +                  'Schur matrix T','Integer',RLABS,'Integer',
     +                  CLABS,80,0,IFAIL)
*
            WRITE (NOUT,*)
            CALL X04DBF('General',' ',N,N,VS,LDVS,'Bracketed','F7.4',
     +                  'Matrix Z of Schur vectors','Integer',RLABS,
     +                  'Integer',CLABS,80,0,IFAIL)
         END IF
*
*        Print workspace information
*
         IF (LWORK.LT.LWKOPT) THEN
            WRITE (NOUT,*)
            WRITE (NOUT,99998) 'Optimum workspace required = ', LWKOPT,
     +        'Workspace provided         = ', LWORK
         END IF
       ELSE
          WRITE (NOUT,*)
          WRITE (NOUT,*) 'NMAX too small'
       END IF
       STOP
*
99999 FORMAT (1X,A,I4)
99998 FORMAT (1X,A,I5,/1X,A,I5)
      END
```

9.2 Program Data

```
F08PNF Example Program Data

  4                                                              :Value of N

(-3.97, -5.04)  (-4.11,  3.70)  (-0.34,  1.01)  ( 1.29, -0.86)
( 0.34, -1.50)  ( 1.52, -0.43)  ( 1.88, -5.38)  ( 3.36,  0.65)
( 3.31, -3.85)  ( 2.50,  3.45)  ( 0.88, -1.08)  ( 0.64, -1.48)
(-1.10,  0.82)  ( 1.81, -1.59)  ( 3.25,  1.33)  ( 1.57, -3.44) :End of matrix A
```

9.3 Program Results

```
 F08PNF Example Program Results

 Schur matrix T
                      1                   2                   3                   4
 1  ( -6.000, -7.000) ( -0.470, -0.212) (  0.044,  0.512) ( -0.910, -0.092)
 2  (  0.000,  0.000) ( -5.000,  2.006) (  0.715, -0.103) ( -0.058,  0.258)
 3  (  0.000,  0.000) (  0.000,  0.000) (  7.998, -0.996) ( -0.223, -1.055)
 4  (  0.000,  0.000) (  0.000,  0.000) (  0.000,  0.000) (  3.002, -4.000)

 Matrix Z of Schur vectors
                      1                   2                   3                   4
 1  ( 0.8457, 0.0000) (-0.3613, 0.1351) (-0.1755, 0.2297) ( 0.1099,-0.2007)
 2  (-0.0177, 0.3036) (-0.3366, 0.4660) ( 0.7228, 0.0000) ( 0.0336, 0.2312)
 3  ( 0.0875, 0.3115) ( 0.6311, 0.0000) ( 0.2871, 0.4999) ( 0.0944,-0.3947)
 4  (-0.0561,-0.2906) (-0.1045,-0.3339) ( 0.2476, 0.0195) ( 0.8534, 0.0000)
```